

## Sortiranje stringova u polju za tekst

U sljedećoj vježbi prikazano je kako, pomoću relacionih operatora i nekoliko funkcija za obradu stringova, sortirati redove u VB polju za tekst. Ovaj program napravljen je kao dorada programa sa prethodnih časova (Qnote) i sadrži komandu Open kojom se otvara postojeći fajl. U meniju File postoji i komanda Sort Text za sortiranje teksta koji se trenutno nalazi u polju zatekst.

Pošto se cijeli sadržaj polja za tekst nalazi u jednom stringu, program mora prvo da podijeli taj string na pojedinačne manje stringove, a ti manji stringovi mogu se zatim sortirati *potprogramom ShellSort*, rutimo, za sortiranje zasnovanom na algoritmu koji je 1959. napravio Donald Shell, Da bi zadatak bio jednostavniji napravljen je modul sa definicijom dinamičkog niza tipa string gdje se u svaki element niza smješta po jedan red iz polja za tekst. U isti standardni modul stavljen je i potprogram ShellSort koji se može pozivati iz bilo koje procedure događaja u programu.

Interesantan dio ovog programa je rutina u kojoj se određuje broj redova u objektu TextBox. U VB ne postoji funkcija koja bi to automatski izračunala. Ideja je bila da program bude u stanju da sortira svako polje za tekst, bez obzira na njegovu veličinu. To se postiže sljedećim programskim kodom u kome se koristi funkcija Chr da bi se pronašao znak za novi red koji se nalazi na kraju svakog reda.

```
'utvrđivanje broja redova u polju za tekst (txtNote)
lineCount% = 0 'u ovoj promenljivoj se čuva ukupan broj redova
charsInFile% = Len(txtNote.Text) 'utvrđivanje ukupnog broja znakova u polju
For i% = 1 To charsInFile% 'pomeranje u polju, znak po znak
    letter$ = Mid(txtNote.Text, i%, 1) 'preneti sledeći znak u promenljivu
                                letter$
    If letter$ = Chr(13) Then 'ako je pronaden znak za novi red (kraj reda!)
        lineCount% = lineCount% + 1 'preći na sledeći red (povećati brojač redova)
        i% = i% + 1 'preskočiti znak linefeed (koji se uvek nalazi iza znaka za
                                novi red)
    End If
Next i%
```

*Originalni kod koji ćemo koristiti (iz knjige) i gdje su komentari na engleskom je:*


```
'determine number of lines in text box object (txtNote)
lineCount% = 0 'this variable holds the total number of lines
charsInFile% = Len(txtNote.Text) 'get total characters in box
For i% = 1 To charsInFile% 'move one char at a time through box
    letter$ = Mid(txtNote.Text, i%, 1) 'put next char in letter$
    If letter$ = Chr$(13) Then 'if carriage ret found (end of line!)
        lineCount% = lineCount% + 1 'go to next line (add to count)
        i% = i% + 1 'skip linefeed char (which always follows cr)
    End If
Next i%
```


Ovom rutinom se u promenljivoj lineCount% vraća broj redova u polju za tekst. Ta vrijednost se koristi da bi se odredila dimenzija dinamičkog niza za pojedinačne redove teksta.

Dobijeni stringovi zatim se prosljeđuju potprogramu ShellSort, koji ga vraća sortirano po abecednom redu. Nakon toga niz se jednostavno kopira nazad u polje za tekst, petljom For.

Forma programa:

## Izvršavanje programa (SortDemo)

- Prvo raspakovati fajl Less12.rar  Less12 (NAPOMENA: fajl Less12.rar poslat mail-om)

- Iskopirati folder  Less12 na C: da bude podfolder  Program Files

1 Kliknite na dugme Open Project na liniji alatki i otvorite projekat SortDemo



Dugme Start


- 2 Kliknite na dugme Start na liniji alatki da bi se program pokrenuo.  
3 U polje za tekst upišite sledeće reči (ili neke druge koje hoćete):

**Zebra**  
**Gorilla**  
**Moon**  
**Banana**  
**Apple**  
**Turtle**

Ne zaboravite da iza poslednje reči pritisnete taster Enter da bi Visual Basic tačno izračunao broj redova.


- 4 U meniju File kliknite na komandu Sort Text.

Tekst koji ste upisali se sortira i ponovo prikazuje u polju na sledeći način:




- 5 U meniju File kliknite na komandu Open i otvorite fajl abc.txt u folderu Less12.

Fajl abc.txt sadrži 36 redova teksta. Svaki red počinje slovom ili brojem (od 1 do 10).



- 6 U meniju File kliknite na komandu Sort Text da bi se sadržaj fajla abc.txt sortirao. Program SortDemo sortira fajl u rastućem redosledu i prikazuje sortirane redove u polju za tekst.



- 7 Pomerajte sadržaj i pregledajte rezultate abecednog sortiranja. Primitičete da, iako je abecedni deo sortiran bez zamerke, postoji neobičan rezultat u numeričkom delu – red koji počinje brojem 10 nalazi se drugi na listi, a ne deseti. To se dogodilo zato što je Visual Basic posmatrao 1 i 0 u broju 10 kao dva zasebna znaka, a ne kao jedan broj. Pošto se poredi ASCII vrednosti stringova sleva udesno, program je izvršio čisto abecedno sortiranje. Ako ovim programom hoćete da sortirate brojeve, treba da brojeve prenesete u numeričke promenljive i poredite ih umesto da koristite funkcije za stringove.

## Analiza programa SortDemo

- 1 U meniju File programa SortDemo kliknite na komandu Exit da biste zaustavili program.
- 2 Otvorite prozor Code (ako nije otvoren) i prikažite programski kôd procedure događaja mnulitemSortText.

Već smo analizirali prvi deo ove procedure događaja u kojem se broje redovi polja za tekst tako što se funkcijom Mid traži znak za prelazak u novi red. U preostalom delu procedure događaja se podešava dužina niza stringova, svaki red teksta se kopira u niz, poziva se potprogram za sortiranje niza i sortirana lista se prikazuje u polju za tekst.

- 3 Pomerite sadržaj prozora Code do drugog dela procedure događaja.

Ekran bi trebalo da izgleda ovako:

```

Project1 - Form1 [Code]
muItenSortText Click
'build an array to hold the text in the text box
ReDim strArray$(lineCount%) 'create array of proper size
curline% = 1
In$ = "" 'use In$ to build lines one character at a time
For i% = 1 To charsInFile% 'loop through text again
    letter$ = Mid(txtNote.Text, i%, 1)
    If letter$ = Chr$(13) Then 'if carriage return found
        curline% = curline% + 1 'increment line count
        i% = i% + 1 'skip linefeed char
        In$ = "" 'clear line and go to next
    Else
        In$ = In$ & letter$ 'add letter to line
        strArray$(curline%) = In$ 'and put in array
    End If
Next i%

'sort array
ShellSort strArray$(), lineCount%

```

Niz strArray\$ je deklarisan u standardnom modulu (SortDemo.bas) koji je takođe deo ovog programa. Naredbom ReDim i promenljivom lineCount% određujem dužinu dinamičkog niza strArray\$. Na taj način se stvara niz sa onolikim brojem elemenata koliko ima redova u polju za tekst (potprogram ShellSort zahteva takav niz). Petljom For i promenljivom In\$ ponovo pretražujem tekst u polju i tražim znak za prelazak u novi red. Zatim kopiram svaki završen red u niz strArray\$. Kada napunim niz, pozivam potprogram ShellSort koji sam prethodno napravio u standardnom modulu SortDemo.bas.

Potprogram ShellSort kod poređenja elemenata u nizu koristi relacioni operator <= i zamenjuje mesta elemenata ako nisu u pravilnom redosledu. Potprogram izgleda ovako:

```

Sub ShellSort(sort$(), numOfElements%)
'Potprogram ShellSort sortira elemente niza sort$()
'u rastućem redosledu i vraća ga u pozivajuću proceduru.
span% = numOfElements% \ 2
Do While span% > 0
    For i% = span% To numOfElements% - 1
        j% = i% - span% + 1
        For j% = (i% - span% + 1) To 1 Step -span%
            If sort$(j%) <= sort$(j% + span%) Then Exit For
            'zameniti mesta elementima koji nisu u pravilnom redosledu
            temp$ = sort$(j%)
            sort$(j%) = sort$(j% + span%)
            sort$(j% + span%) = temp$
        Next j%
    Next i%
    span% = span% \ 2
Loop
End Sub

```

Kod ovog načina sortiranja se glavna lista elemenata stalno deli na podliste koje su svaki put upola manje. Zatim se porede početak i kraj podliste da bi se utvrdilo da li su u pravilnom redosledu. Ako početak i kraj nisu u pravilnom redosledu, zamenjuju im se mesta. Krajnji rezultat je lista sortirana u rastućem abecednom redosledu. Da bi se postigao opadajući redosled, samo treba zameniti relacioni operator (umesto <= staviti >=).

D. zadatak:

Napraviti novi fajl abv.txt i sortirati ga ovim programom. U njemu koristiti riječi iz našeg jezika (npr. S je za školu, itd.).